
 1219 EMULATOR OPERATION

Works pretty much like a real machine. Click on registers , type in octal values into registers and click 
start.. Well, we were really punching in binary, but we were thhinking in octal

To manually store a program:

1. click on Function repeate button to light red

2. click on RUN button to change ledgend to "Op Step"

3. Enter first address to be stored into P register

4. put octal 44 (Store AL) in the F (Function) register

5. Put  word to be stored in AL (Accumulator Lower) register

6. Click start , word will be stored , repeat steps 5& 6 until all words are in

To manually view memory contents:

1. click on Function repeate button to light red

2. click on RUN button to change ledgend to "Op Step"

3. Enter first address to be read into P register

4. put a 12 (Enter AL) in the F (Function) registerd

5. Click start , word will be viewed in AL, repeat this step fot each word

To load or store a memory Images  

NOTE: Memory Image”0” is loaded at emulator start

To load a saved image:

Clicking  on Load with  1 in “P” register will load  a memory image (32K)as referenced by “AL” register 
bits ( 0-2) 

To Save a memory image:

Clicking  on Load with  2 in “P” register will save memory image (32K) as referenced by “AL” register 
bits ( 0-2)

UNIVAC 1219 SOFTWARE

At this time there is not much software. The “memory  image 0” sent with the 1219 emulator 
contains the following:

1. A program residing at octal 1000 that prints out HELLO WORLD over and over until 
stooped by selecting stop key 0

2. A program residing at octal 2000 that inputs a characters from the keyboard and displays
the octal code of the key pressed in register AL

3. Paper tape Bootstrap at octal 500  which unconditionally loads 147(8) at address 540 and 
jumps to 540 this should be a loader program able to read address from tape and load an 
object program and  verify the checksum . 

4. Paper tape loader at address 540

5. A keyboard inspect and change program at octal 70000with the following commands:

• I (inspect) type 6 digit address or shorter terminated by a space

• N (next) inspects next address

• L (last) goes back one address

• C (change) inputs 6 digit octal number to be stored at last address inspected

• T  (terminal) inputs 6 digit terminal address

• M (mask) inputs 6 digit mask

• D (data) inputs 6 digit data word

• D (data) inputs 6 digit data word

• CTRL+D (dump)  Dumps contents of memory from inital to terminal

• CTRL+S  (search) Searches form initial to terminal for data subject to mask  
(must match where bits are set in mask)

1532 INPUT-OUTPUT CONSOLE  NOTES

1. Console defaults to channel 0; No need to select Channel unless program uses 
something else. Almost all 1532's  were on  channel 0, as it had the lowest I/O 
priority.    

2. Punch feature not implemented at this time

3. Put file name in File window before starting the bootstrap or loader. Failing to do 
so will cause  message “Mount File and select Reader”; Clear message put file 
name in file window and  select reader. The assembler “.76” programs load with 
the loader at octal address 540.

4. I added a check box to inhibit carriage returns. Modern computer equipment use a
linefeed character to both  linefeed and character return.. Legacy programs, using 
both, would skip lines

1219 ASSEMBLER NOTES

The assembler program started many years as an assembler I wrote on a Radio Shack Color Computer  for the 
AN/UYK-20 computer; which was the standard US Navy minicomputer at the time. Not having Disc I/O all all 
source code / object listing had to fit in memory. Source was entered as data statements in the program. Later when 
I bought my first P.C. (a Packard bell 80286 machine with 40 Mb HD) I converted it to GW basic. About the time 
all the 1219's went away ,I modified it to assemble 1219 instructions. Now it is running under Q basic 64. Maybe 
some day I will port it to Visual basic. The code is UGLY. To save memory, I used all one or two character variable 
names, a gazillion statements per line and no comments.

To run assembler start program and enter the name  of the source  program , in upper case, containing your 
program's  source code in TRIM neumonics. Enter only name portion as assembler adds the .txt . The assembler 
writes the output files in the same directory as it is run from. The output  name .PRG is the assembly listing. The 
name.76 is a bioctal tape image that can be loaded into the computer by putting it's name in the file window of the 
1532 I/O console and starting the computer at 540 (8)   .  The assembler was assembled with QBASIC and 
must be run in a DOS emulator on late model computers.( I use DosBox)

 CODING CONVENTIONS:

• Labels must start in column 1 and be preceded by a >
     >LABLE1 ENTAL LOK+2

•  All numbers are assumed to be decimal values except:
      &100  octal 100 (decimal 64)
      &HABC hexadecimal ABC (decimal 2748)
 

 ASSEMBLER DIRECTIVES AND PSEUDO OPERATIONS:

• ORG  Sets beginning address at which to assemble the following instructions. multiple ORG 
directives are permitted.

•  END  Terminates assembly.

• SADD Sets Program starting address. If present assemler will produce self starting tape.

•  RES Skips a number of addresses.

• DATA Enters data into the current address.

•  BCW Same as  DATA but lets you use BCW to make your listing easier to read.  

•   AS  (ASCII string) Packs the message that follow into memory two ASCII  characters per 
address.“[“= RETURN ; “]” = LINEFEED . Bit 9 is set in last character

•  AW   Puts the text that follows into memory one ASCII  character per memory address.“[“ = 
RETURN ; “]” = LINEFEED. Bit 9 is set in last character.

• EVEN  Causes the next instruction (usually data) to be  assembled an even address.
The 1219 computer requires double length operands to be at even addresses
as it forces bit 0 set to retreive the most siginifant half.

• EQU  Assigns the value of its operand to its label.

•   ;    The rest of  line is taken as a remark.

• LOK Holds the value of  it's address. Do not put space before offset .LOK+4 correct LOK  +4 wrong

SAMPLE ASSEMBLER SOURCE LISTING

;SAMPLE BOOTSTRAP FOR UNIVAC 1219 COMPUTER 
;COMPATABLE WITH TARTAR LOADER 
;PROGRAM UNCONDITIONALY LOADS 147 OCTAL WORDS (LOADER) 
;ON BIOCTAL PAPER TAPE STARTING AT OCTAL ADDRESS 436
;THEN JUMPS TO OCTAL 540 (FIRST 2 WORDS ARE HEADER)
;TAPE FORMAT: 76=> BIOCTAL TAPE, IIIII =>INITIAL, TTTTT=> TERMINAL
;76 (BIOCTAL TAPE ID CODE)
;II (MSD INITIAL ADDRESS)
;II
;IT
;TT
;TT (LSD OF FINAL ADDRESS)
;PP (FIRST 6 BITS OF FIRST INSTRUCTION)
;PP
;PP (LAST 6 BITS OF FIRST INSTRUCTION)
;** REMEMBER BOOTSTRAP IGNORES HEADER DATA**
;
;
ORG &O500 ; ADDRESS TO START ASSEMBLY
>TARBOOT SIL 0
ENTSR &O10
ENTICR 1
EXF 00  ; TURN ON READER
BCW READ
BCW READ 
ENTBK 0 ;SET FRAME COUNT
>NXTWD STRB 3
ENTALK 2
STRAL 4
ENTALK 0
>INPUT IN 0 ;GET FRAME
BCW 2
BCW 2
SKPIIN 0 ; WAIT FOR DATA
JP LOK-1
LSHAL 6  ; SHIFT UP 6 BITS
SLSET 2 
JPALNZ LOK+3
ENTAU 3
JPAUZ INPUT
ISK 4  ;THREE FRAMES?
JP INPUT ; NO GET MORE
STRALB &O536 ;YES STORE WORD 
BSK WDCT ;DONE YET?
JP NXTWD ;NO GET NEXT
JP &O540  ;yES JP TO LOADER
>WDCT DATA &O147 ;WORDS TO LOAD
>READ DATA &O151 ; CODE TO TURN ON READER
END

SAMPLE ASSEMBLER OBJECT LISTING

TARTAR.PRG
;SAMPLE BOOTSTRAP FOR UNIVAC 1219 cOMPUTER  
;COMPATABLE WITH TARTAR LOADER  
;PROGRAM UNCONDITIONALY LOADS 147 OCTAL WORDS (LOADER)  
;ON BIOCTAL PAPER TAPE STARTING AT OCTAL ADDRESS 436 
;THEN JUMPS TO OCTAL 540 (FIRST 2 WORDS ARE HEADER) 
;TAPE FORMAT: 76=> BIOCTAL TAPE, IIIII =>INITIAL, TTTTT=> TERMINAL 
;76 (BIOCTAL TAPE ID CODE) 
;II (MSD INITIAL ADDRESS) 
;II 
;IT 
;TT 
;TT (LSD OF FINAL ADDRESS) 
;PP (FIRST 6 BITS OF FIRST INSTRUCTION) 
;PP 
;PP (LAST 6 BITS OF FIRST INSTRUCTION) 
;** REMEMBER BOOTSTRAP IGNORES HEADER DATA** 
; 
; 
                                 ORG &O500 ; ADDRESS TO START ASSEMBLY 
TARBOOT   000500 503400          SIL 0 
          000501 507310          ENTSR &O10 
          000502 507201          ENTICR 1 
          000503 501300          EXF 00  ; TURN ON READER 
          000504 000534          BCW READ 
          000505 000534          BCW READ  
          000506 360000          ENTBK 0 ;SET FRAME COUNT 
NXTWD     000507 420003          STRB 3 
          000510 700002          ENTALK 2 
          000511 440004          STRAL 4 
          000512 700000          ENTALK 0 
INPUT     000513 501100          IN 0 ;GET FRAME 
          000514 000002          BCW 2 
          000515 000002          BCW 2 
          000516 502100          SKPIIN 0 ; WAIT FOR DATA 
          000517 340516          JP LOK-1 
          000520 504606          LSHAL 6  ; SHIFT UP 6 BITS 
          000521 510002          SLSET 2  
          000522 630525          JPALNZ LOK+3 
          000523 100003          ENTAU 3 
          000524 600513          JPAUZ INPUT 
          000525 570004          ISK 4  ;THREE FRAMES? 
          000526 340513          JP INPUT ; NO GET MORE 
          000527 450536          STRALB &O536 ;YES STORE WORD  
          000530 560533          BSK WDCT ;DONE YET? 
          000531 340507          JP NXTWD ;NO GET NEXT 
          000532 340540          JP &O540  ;yES JP TO LOADER 
WDCT      000533 000147          DATA &O147 ;WORDS TO LOAD 
READ      000534 000151          DATA &O151 ; CODE TO TURN ON READER 

    LABEL TABLE

TARBOOT       000500
INPUT         000513
NXTWD         000507
READ          000534
WDCT          000533

